

NAG Library Routine Document

G05SKF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

G05SKF generates a vector of pseudorandom numbers taken from a Normal (Gaussian) distribution with mean μ and variance σ^2 .

2 Specification

```
SUBROUTINE G05SKF (N, XMU, VAR, STATE, X, IFAIL)
  INTEGER          N, STATE(*), IFAIL
  REAL (KIND=nag_wp) XMU, VAR, X(N)
```

3 Description

The distribution has PDF (probability distribution function)

$$f(x) = \frac{1}{\sigma\sqrt{2\pi}} \exp\left(-\frac{(x-\mu)^2}{2\sigma^2}\right).$$

G05SKF uses the algorithm of Wichura (1988).

One of the initialization routines G05KFF (for a repeatable sequence if computed sequentially) or G05KGF (for a non-repeatable sequence) must be called prior to the first call to G05SKF.

4 References

- Kendall M G and Stuart A (1969) *The Advanced Theory of Statistics (Volume 1)* (3rd Edition) Griffin
- Knuth D E (1981) *The Art of Computer Programming (Volume 2)* (2nd Edition) Addison–Wesley
- Wichura (1988) Algorithm AS 241: the percentage points of the Normal distribution *Appl. Statist.* **37** 477–484

5 Parameters

- | | | |
|----|--|--------------|
| 1: | N – INTEGER | <i>Input</i> |
| | <i>On entry:</i> n , the number of pseudorandom numbers to be generated. | |
| | <i>Constraint:</i> $N \geq 0$. | |
| 2: | XMU – REAL (KIND=nag_wp) | <i>Input</i> |
| | <i>On entry:</i> μ , the mean of the distribution. | |
| 3: | VAR – REAL (KIND=nag_wp) | <i>Input</i> |
| | <i>On entry:</i> σ^2 , the variance of the distribution. | |
| | <i>Constraint:</i> $VAR \geq 0.0$. | |

4: STATE(*) – INTEGER array Communication Array

Note: the actual argument supplied **must** be the array STATE supplied to the initialization routines G05KFF or G05KGF.

On entry: contains information on the selected base generator and its current state.

On exit: contains updated information on the state of the generator.

5: X(N) – REAL (KIND=nag_wp) array Output

On exit: the n pseudorandom numbers from the specified Normal distribution.

6: IFAIL – INTEGER Input/Output

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, $N = \langle value \rangle$.

Constraint: $N \geq 0$.

IFAIL = 3

On entry, $VAR = \langle value \rangle$.

Constraint: $VAR \geq 0.0$.

IFAIL = 4

On entry, STATE vector has been corrupted or not initialized.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.8 in the Essential Introduction for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.7 in the Essential Introduction for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.6 in the Essential Introduction for further information.

7 Accuracy

Not applicable.

8 Parallelism and Performance

G05SKF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

None.

10 Example

This example prints five pseudorandom numbers from a Normal distribution with mean 1.0 and variance 1.5, generated by a single call to G05SKF, after initialization by G05KFF.

10.1 Program Text

```

Program g05skfe

!      G05SKF Example Program Text

!      Mark 25 Release. NAG Copyright 2014.

!      .. Use Statements ..
Use nag_library, Only: g05kff, g05skf, nag_wp
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: lseed = 1, nin = 5, nout = 6
!      .. Local Scalars ..
Real (Kind=nag_wp)         :: var, xmu
Integer                    :: genid, ifail, lstate, n, subid
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: x(:)
Integer                    :: seed(lseed)
Integer, Allocatable       :: state(:)
!      .. Executable Statements ..
Write (nout,*) 'G05SKF Example Program Results'
Write (nout,*)

!      Skip heading in data file
Read (nin,*)

!      Read in the base generator information and seed
Read (nin,*) genid, subid, seed(1)

!      Initial call to initialiser to get size of STATE array
lstate = 0
Allocate (state(lstate))
ifail = 0
Call g05kff(genid,subid,seed,lseed,state,lstate,ifail)

!      Reallocate STATE
Deallocate (state)
Allocate (state(lstate))

!      Initialize the generator to a repeatable sequence
ifail = 0
Call g05kff(genid,subid,seed,lseed,state,lstate,ifail)

```

```
!      Read in sample size
      Read (nin,*) n

      Allocate (x(n))

!      Read in the distribution parameters
      Read (nin,*) xmu, var

!      Generate the variates
      ifail = 0
      Call g05skf(n,xmu,var,state,x,ifail)

!      Display the variates
      Write (nout,99999) x(1:n)

99999 Format (1X,F10.4)
      End Program g05skfe
```

10.2 Program Data

```
G05SKF Example Program Data
1 1 1762543      :: GENID,SUBID,SEED(1)
5              :: N
1.0 1.5        :: XMU,VAR
```

10.3 Program Results

```
G05SKF Example Program Results
```

```
 1.4272
-0.5254
 1.8109
 2.0232
-0.5380
```
